
Tamnoon
    

A simplicity-first web framework for Elixir.
Tamnoon is a web framework designed to offer a simple, minimal, and approachable way to build web applications with Elixir. It emphasizes rapid development by eliminating boilerplate, allowing you to focus on building features rather than configuring setup. Inspired by modern frontend tools such as React and Redux, Tamnoon embraces a component-based, functional architecture tailored for Elixir developers.

  
    
  
  Highlight Features


	Zero configuration required – Get started instantly with no setup or boilerplate.

	Built-in PubSub – Simple, native publish-subscribe messaging without extra dependencies.

	Pure Elixir – Build full-featured interactive web apps without writing any JavaScript.


To get started, check out the Overview guide.

  
    
  
  Installation


The package can be installed by adding tamnoon to your list of dependencies in mix.exs:
def deps do
  [
    {:tamnoon, "~> 1.0.0-rc.1"}
  ]
end
The docs can be found at https://hexdocs.pm/tamnoon.



  

    
Changelog
    


  
    
  
  v1.0.0-rc.1 (30.07.25)


	Added DOM Actions: Methods can now return a third argument of actions, which are used to manipulate the DOM in ways that are impossible or convoluted to achieve with the state.

	Revamped the documentation.

	Methods will now return a tuple of the form {}, {diffs}, {diffs, actions} or {diffs, actions, new_state}. For the first two, diffs will automatically be merged into the state.

	Changed Tamnoon.Methods.subbed_channels/0 to a regular function instead of a method.


Minor changes:
	Removed Tamnoon.MethodManager's diff/2 function.

	Changed debug mode to be able to log only the req or state, and changed the debug messages formatting.

	Changed the default router to use plug's Plug.Static.

	Renamed the driver script to tamnoon_driver.js (was ws_connect.js).

	Changed methods to accept the "value" key for the value instead of "val".

	Changed sub and unsub to accept the channel name via the "channel" key.

	Fixed a bug where after setting an element's disabled attribute to a state value, changes to it wouldn't re-enable the element properly.

	Setting an element's class to a state value will now not override Tamnoon's listener classes.

	Silenced the unused variable "state" warnings for methods.

	Changed the root component that gets generated with Mix.Tasks.Tamnoon.OverrideRoot to not try rendering an app.html.heex component by default.

	Fixed a bug causing <a> elements to not send their value when triggering a method.

	Fixed a bug where having a hash (#) in the URL would cause the socket connection to error out.

	Renamed the timeout parameter of Tamnoon.MethodManager.trigger_method/3 to timeout_ms.



  
    
  
  v1.0.0-a.5 (13.06.25)


	Made Tamnoon HEEx values invertable: for example, assigning @not-some_value to an attribute will turn it true when some_value is false, and vice versa.

	Added the tamnoon.setup task, which runs all recommended tasks for starting a new Tamnoon app.

	Removed the app-container div from the default root, and fixed the indentation in its HTML.

	Fixed a bug where Tamnoon.MethodManager.defmethod/2 caused a no match found error.


Minor changes:
	Fixed a bug where live reload did not reload some changes for the client that triggered it.

	Removed the (empty) documentation page for the default root component.

	Altered the "no method found" error message to be clearer.

	Made the Tamnoon.Component documentation more concise.



  
    
  
  v1.0.0-a.4 (05.06.25)


	Added live reload. When not disabled, Tamnoon will automatically recompile when new connections are made (including existing connections refreshing the page).

	Changed method modules such that now multiple modules are used instead of a singular one. Deprecated the __using__ macro of Tamnoon.Methods because of this too.

	Added the tamnoon.override_root Mix task. The task generates a router and a root component for you.

	Added debug mode. Enabling it will log the payload and current state whenever a method is triggered.

	Added Tamnoon.MethodManager.diff. (Removed in v1.0.0-rc.1)

	Added Tamnoon.MethodManager.trigger_method/3.

	Fixed bugs where unrelated HTML classes interfere with Tamnoon classes.

	Added support for setting initial_state as a function, allowing the initial state to be recomputed for every client.

	Silenced the unused variable "req" warnings for methods.



  
    
  
  v1.0.0-a.3 (25.07.24)


	Added support for running Tamnoon over HTTPS.

	Added support for Mix Releases (via Tamnoon.make_release/1).

	Updated the client script to try reconnecting to the server if it disconnects.

	Updated the client script to keep a copy of the state which will be sent to the server on a reconnect. Note: this also means that string keys in the state are now not supported, as the state will have its keys converted to atoms on the reconnect.


Minor changes:
	Added Tamnoon.Methods.tmnn_set_state/2.

	Added deployment guide (WIP)



  
    
  
  v1.0.0-a.2 (29.06.24)


	Added implicit events: setting an input event attribute (such as onchange) to @method-key will make it fire the method with the specified key. The "pub" method can be triggered like so: @pub-[channel]-[method]-[key (optional)].

	A component with a Tamnoon variable inside it (<p>@val</p>) will now set the inner text of the element instead of the inner HTML. The inner HTML can be set by prefixing the variable name with "raw-" like so: <p>@raw-val</p>.

	Added the mix tamnoon.make_dirs task that creates a "lib/components" directory and a "tamnoon_out" directory.

	Added Tamnoon.Compiler.escape_html/1 and made it available as <%= h.(content) %> inside components.

	Completely rewrote the guides, the readme, and revamped some of the existing module documentation.


Minor changes:
	Changed the default port to 8000 from 4000.

	Fixed "imported function conflicts with local function" when invoking a method handler inside the methods module.

	Fixed the client-side handling of updates for elements with the "value" attribute not functioning properly.

	Changed the WebSocket script in the root layout to connect to the WebSocket at the current URL (instead of localhost:8000/ws).

	Changed Tamnoon.Compiler.build_from_root/1 to no longer accept a WebSocket address.



  
    
  
  v1.0.0-a.1 (26.06.24)


	Added the sync method (Tamnoon.Methods.tmnn_sync/2) that returns the current state to the client. By default, it is invoked when the WebSocket connection is opened.

	Added the keep_alive method (Tamnoon.Methods.tmnn_keep_alive/2) that is invoked every 55 seconds by the client in order to prevent idle timeouts.

	Moved the WebSocket connection script to a separate file, and updated the default router (Tamnoon.Router) to serve a default root layout.


Minor changes:
	Changed the console message sent when Tamnoon starts to include the full address.

	Changed Tamnoon.SocketHandler.init/2 so connections have a 120 second idle timeout (instead of 60).

	Added Tamnoon.Compiler.render_component_dyn/1.

	Changed Tamnoon.Compiler.render_component/3 so using <%= r.(args..) %> will call Tamnoon.Compiler.render_component_dyn/1 instead of Tamnoon.Compiler.render_component/1, to allow passing in multiple arguments.

	Added (temporary) documentation.



  
    
  
  v1.0.0-a.0 (25.06.24)


Added support for HEEx components and changed the default router to display the root page by default.

  
    
  
  v0.1.1


Very minor changes.

  
    
  
  v0.1.0


First release.



  

    
Overview
    

Tamnoon is a modern Elixir framework for building frontend web applications, designed to streamline and simplify development, and stay true to the principles of the language. It is built around the following core goals:
	Simplicity first - Tamnoon adopts a clear, component-driven structure that reduces complexity and helps developers turn ideas into features with minimal friction.

	Zero boilerplate - By removing unnecessary setup and configuration, Tamnoon enables fast, efficient development with clean and focused code.

	Functional by design - Built to align with Elixir’s functional nature, Tamnoon encourages a pure, declarative style that fits naturally into the language.



  
    
  
  Guides


The following guides will walk you through the fundamentals of working with Tamnoon. A basic understanding of HTML is assumed, and while not required, familiarity with tools like React or Redux may help you get comfortable more quickly.
	Getting Started - Installing Tamnoon and initializing a project.

	Components - An overview of components, including how to define, structure, and utilize them within your application.

	Methods - Tamnoon's approach to defining behaviors and handling logic within components.

	PubSub - Tamnoon's built-in support for publish-subscribe communication, enabling real-time updates and coordination between multiple connected clients.

	Tamnoon HEEx - A guide to Tamnoon's templating language for components.

	DOM Actions - Mechanisms for directly interacting with the DOM outside of state management, to address edge cases and add low-level control.



  
    
  
  Sample Apps


If you learn best by example or just want a reference point, here are some sample applications built with Tamnoon:
	Chatroom: A chat app with rooms and users, demonstrating Tamnoon's PubSub functionality.

	Actions Test: A demo app for Actions.


Note: This section is WIP and will be expanded in the future.



  

    
Getting Started
    

To begin using Tamnoon, create a new Elixir project with a supervision tree:
$ mix new [NAME] --sup

Next, add Tamnoon to your deps in mix.exs:
  defp deps do
    [
      # Other dependencies...
      {:tamnoon, "~> 1.0.0-rc.1"}
    ]
  end
Then, run the setup command to initialize the project:
$ mix tamnoon.setup

This command will guide you to update your application’s start_link function to include Tamnoon. Follow the instructions provided in the terminal, and your project will be ready to go.

  
    
  
  What Just Happened?


The mix tamnoon.setup command adds a few essential files and directories to your project structure. Here's what each one is for:
	tamnoon_out/ - A directory that contains the compiled HTML file to be served to the client.

	lib/components/ - The directory where your app's components are defined.

	lib/components/root.ex - The root component of your application, serving as the entry point for your UI.

	lib/router.ex - A Plug.Router module that handles serving static files for your Tamnoon app.



  
    
  
  Running The App


Once the setup is complete, you can start your application using the following command:
$ mix run --no-halt

If everything is set up correctly, you should see the following output in your terminal:
10:34:54.305 [info] Tamnoon listening on http://localhost:8000..

At this point, your app is up and running. You can now proceed to the Components guide to begin building your interface.



  

    
Components
    

In Tamnoon, the user interface is built using components - reusable building blocks written in Tamnoon HEEx, Tamnoon's templating language.
Tamnoon HEEx is an extension of HEEx (HTML + Elixir), with additional features tailored specifically for Tamnoon, which will be covered in the Tamnoon HEEx guide.
In practice, components can be created in one of two ways:
	Elixir modules that implement the Tamnoon.Component behaviour:

defmodule MyApp.Components.MyComponent do
  @behaviour Tamnoon.Component

  @impl true
  def heex do
    ~s"""
    <h1>Hello World!</h1>
    """
  end
end
	.html.heex files containing Tamnoon HEEx markup directly:

<h1>Hello World!</h1>

  
    
  
  Injecting Code Into Components


While components can be written as plain HTML, you'll often need to introduce dynamic behavior. This is where EEx (Embedded Elixir) comes into play.
Tamnoon HEEx supports standard EEx syntax, allowing you to embed Elixir code directly within your components using <% %> and <%= %> tags:
	<%= %> evaluates and renders the result.

	<% %> evaluates the code but does not render any output.


Example
<% time = Time.utc_now() %>
<p>
  <%= if Time.before?(time, ~T[12:00:00]), do: "Good morning", else: "Good
  day"%> world!
</p>
This will render either Good morning world! or Good day world! depending on the current time.
Note that this code runs once when the component is rendered, not on every DOM update.
The h.() helper
Inside EEx tags in components, you can use the h.()helper function, which will escape any HTML inside a string (see Tamnoon.Compiler.escape_html/1).


  
    
  
  Rendering Other Components


Tamnoon allows you to render components from within another using the r.() helper, which internally calls Tamnoon.Compiler.render_component_dyn/1.
You can use this syntax inside Tamnoon HEEx templates:
<%= r.(MyApp.Components.MyComponent) %> <%= r.("my_component.html.heex") %>
Both forms will inject the output of the specified component into the current one:
	The first renders a component defined as an Elixir module.

	The second renders a component defined in a .html.heex file.


This makes it easy to compose complex UIs from smaller, reusable pieces.

  
    
  
  Using Assigns


Assigns allow you to pass data into components - similar to properties or props in other frameworks. This is especially useful for customizing styling, content, or behavior.
To pass assigns, provide a two-element list: the component and a map of key-value pairs:
<%= r.([MyApp.Components.MyButton, %{color: "primary" }]) %>
Inside the component, these values can be accessed using the @ sigil, just like in standard HEEx:
<button class="<%= "btn-" <> @color %>">Click Me!</button>
In this example, the rendered output would be:
<button class="btn-primary">Click Me!</button>



  

    
Methods
    

This guide will cover methods - the main way to add interactions to your app. Methods are functions that are used to change your app's state.

  
    
  
  The State


A Tamnoon app maintains a state for each client - a map of key-value pairs that can be read and updated in response to user interactions. This state is initialized as part of your Tamnoon options:
def start(_type, _args) do
  children = [
    {Tamnoon, [[
      initial_state: %{
        message: "Hello World!",
        button_hidden: true,
        username: "user"
      },
      # Other options...
    ]]}
  ]
  ...
end
Note: State keys must be atoms - not strings.

To interact with the application state, you'll need to use Tamnoon HEEx, which is covered in the next guide.

  
    
  
  Methods


Methods are Elixir functions triggered by user interactions in the UI. They allow you to perform logic and update the application state. Methods are defined using the Tamnoon.MethodManager.defmethod/2 macro inside designated methods modules.
Defining a Methods Module
To define a methods module, create a new Elixir module - typically placed under lib/methods/ - and register it in your Tamnoon options:
def start(_type, _args) do
  children = [
    {Tamnoon, [[
      methods_modules: [
        MyApp.Methods.MyMethodsModule
      ]
      # Other options...
    ]]}
  ]
  ...
end
Defining a Method
Inside your methods module, import Tamnoon.MethodManager and use the Tamnoon.MethodManager.defmethod/2 macro to define methods:
defmodule MyApp.Methods.MyMethodsModule do
  import Tamnoon.MethodManager

  defmethod :ping do
    IO.inspect("pong")

    {%{}}
  end
end
In this example:
	:ping is the name of the method.

	The method prints "pong" to the console.

	It returns a tuple {diffs}, where diffs is a map of state changes.



  
    
  
  Method Parameters


The Tamnoon.MethodManager.defmethod/2 macro provides your method with two arguments:
	state: The current application state as a map.

	req: A map containing metadata about the method invocation:
	:value: The value of the element that triggered the method.
	:key: A custom key provided when invoking the method (optional, only included if one is provided).
	:element: The raw HTML string of the invoking element.




  
    
  
  Method Return Values


A method must return one of the following:
	An empty tuple {}.

	A single-element tuple {diffs}.

	A two-element tuple {diffs, actions}.

	A three-element tuple {diffs, actions, new_state}.


The second element, actions, will be explained in the Actions guide.

	diffs: A map of state changes that will be sent to the client.

	new_state: The updated full state after the method is executed. If not provided, diffs will be merged into the state automatically.


Example
The snippet below is a full example of a methods module containing a method :increment which increments a :counter value:
defmodule MyApp.Methods.CounterMethods do
  import Tamnoon.MethodManager

  defmethod :increment do
    current_count = Map.get(state, :counter)

    {%{counter: current_count + 1}}
  end
end

  
    
  
  Calling Methods From Other Methods


Tamnoon allows you to call one method from another, which is useful for composing logic and avoiding duplication.
There are two main ways to do this:
	Manual invocation

You can directly invoke a method from within another by calling the underlying function it defines.
	Built-in methods can be called using Tamnoon.Methods.<function_name>.

	Custom methods defined using Tamnoon.MethodManager.defmethod/2 are available as regular functions with the prefix tmnn_ in the defining module.


The result of the invoked method must be manually handled and returned from the calling method.
Example
In the example below, triggering the :ping_and_color method will update the :message to "pong" and set its color to "green":
defmethod :ping do
  {%{message: "pong"}}
end

defmethod :ping_and_color do
  {%{message: message}} = tmnn_ping(%{}, state)

  {%{message_color: "green", message: message}}
end
Here:
tmnn_ping/2 is called with an empty req and the current state.
Its result is destructured and merged with additional changes before returning.
	Using trigger_method/3

Tamnoon also provides the Tamnoon.MethodManager.trigger_method/3 function, which allows you to invoke a method by name from within another method.
trigger_method(method_name :: atom, req :: map, timeout_ms :: integer)
	method_name - The name of the method to trigger (as an atom).

	req - A map that will be passed as the method’s req argument.

	timeout_ms - (Optional) Time in milliseconds before the method runs. Omit or use 0 for immediate execution.


Key Difference
Unlike manual invocation, trigger_method/3 automatically sends the result of the called method to the client. This means you don't need to merge or return the result yourself - the response is handled for you.
Example
defmethod :ping do
  {%{message: "pong"}}
end

defmethod :ping_and_color do
  Tamnoon.MethodManager.trigger_method(:ping, %{}, 1000)

  {%{message_color: "green"}}
end
In this example:
	When :ping_and_color is triggered, the :message_color is immediately updated to "green".

	After 1 second, the :ping method is triggered, updating the :message to "pong".


This is especially useful for delayed interactions, animations, or chaining state updates over time.

  
    
  
  Built-In Methods


Tamnoon includes a set of built-in methods available for use in your application. Their documentation is located in the Tamnoon.Methods module.



  

    
PubSub
    

PubSub (Publish-Subscribe) is a messaging pattern where clients publish messages to channels, and other clients subscribed to those channels receive and react to the messages in real time. Tamnoon provides built-in PubSub functionality, enabling communication between multiple clients.
In Tamnoon's case:
	Clients publish method calls to channels.

	Other clients subscribed to those channels will receive and execute the method calls.

	Clients can subscribe and unsubscribe from channels dynamically.


This allows for real-time interaction between clients - ideal for collaborative apps, chat rooms, live dashboards, and more.
Example
If a client publishes an :add_message method to the channel "room_123", all clients subscribed to "room_123" will receive and execute the :add_message method.

  
    
  
  Channels


When publishing method calls, they are sent to a specific channel. Clients can dynamically join or leave channels using the built-in :sub and :unsub methods, and view their current subscribed channels using Tamnoon.Methods.subbed_channels/0.
	:sub subscribes the client to a channel.

	:unsub unsubscribes the client from a channel.

	Both methods expect a channel name passed in as req["channel"].


Channels are automatically created when a client attempts to subscribe to one that doesn't exist.
The clients channel
All clients are automatically subscribed to a special channel named "clients". This channel is non-leavable.

Example
defmethod :switch_room do
  target_room_id = req["value"]

  Tamnoon.Methods.unsub(%{"channel" => "room_\#{state[:current_room_id]}"}, state)
  Tamnoon.Methods.sub(%{"channel" => "room_\#{target_room_id}"}, state)

  {%{current_room_id: target_room_id}}
end
In the example above:
	The client leaves their current room channel (e.g., "room_1").

	Then joins the new room channel (e.g., "room_2").

	The client's state is updated to reflect the new current_room_id.



  
    
  
  Publishing


In Tamnoon, clients communicate with each other by publishing method calls. These calls are broadcast to a specified channel, where all clients subscribed to that channel (including the sender) will receive and execute the method as if they triggered it themselves.
To publish a method call, use the built-in :pub method, which accepts:
	"channel" - the name of the channel to publish to.

	"action" - a map containing:
	"method" - the method name to call.

	Any additional data to pass along via req.




Example
defmethod :send_message do
  %{
    current_message: current_message,
    current_room_id: current_room_id
  } = state

  Tamnoon.Methods.pub(%{
    "channel" => "room_#{current_room_id}",
    "action" => %{
      "method" => "add_message",
      "message" => current_message
    }
  }, state)

  {%{current_message: ""}}
end
What this does:
	Publishes a call to the :add_message method on the channel for the current room.
	All clients in that room (including the sender) will receive and run :add_message.

	The message content is sent as req["message"].



	Clears the :current_message state field on the sender's side only (clean-up).


This is a common pattern in Tamnoon:
	One method (e.g., :send_message) handles logic only the publisher needs (validation, cleanup, etc.).

	It then publishes another method (e.g., :add_message) that contains logic for all clients in the channel.






  

    
Tamnoon HEEx
    

Tamnoon HEEx extends standard HTML + EEx by introducing reactivity and interactivity. It allows your UI to respond to events and manipulate the application’s state directly from components.

  
    
  
  Reading From the State


Within components, you can access state values using the @ symbol. Tamnoon HEEx allows you to inject these values directly into your markup. Here are some examples:
<p>@message</p>
<!-- Renders: <p>Hello World!</p> -->

<button hidden=@button_hidden>Can you see me?</button>
<!-- Renders a hidden button if @button_hidden is true -->

<p>Welcome back, <span>@username</span>!</p>
<!-- Renders: <p>Welcome back, user!</p> -->

  
    
  
  Negation


You can prefix any state value with not- in order to negate its boolean value. This is especially useful for toggling visibility, disabling elements, or applying conditional logic in your UI.
Example
<button hidden=@not-show_button>Can you see me?</button>
In the example above:
	When :show_button is true, @not-show_button becomes false, and the button is shown.

	When :show_button is false, @not-show_button becomes true, and the button is hidden.



  
    
  
  Injecting HTML


By default, state values are HTML-escaped to prevent injection vulnerabilities. However, if you intentionally want to render raw HTML from your state, you can opt in by prefixing the key with raw-. For example:
<div>@raw-my_unescaped_html</div>
This will render any HTML code that :my_unescaped_html contains.
Warning: use this feature with caution to avoid introducing security risks such as XSS (cross-site scripting).
Mixing state values with static text
When used as an element's content, state values must be used as its entire content. They cannot be mixed directly with static text.
To combine dynamic values with static content, wrap the dynamic part in an inline element such as <span> - see the third example above.


  
    
  
  Interacting With the State


Tamnoon HEEx allows you to trigger methods directly from user interactions with components. This functionality comes in several main forms:
1. Simple Trigger
By setting an element's event handler (such as onclick, onchange, etc.) to a method name prefixed with @, the method will be invoked when the event fires. For example:
<button onclick=@ping>Ping!</button>
Clicking the button will invoke the :ping method.
2. Triggering With Arguments
You can also pass arguments to methods directly from the element. This can be done in two ways:
a. Passing a key
A key can be passed by appending it with a dash to the method name:
<input onchange=@update-name placeholder="Enter your name.." />
This will invoke the :update method with the key "name", available as req["key"] in the method.
Note: For the :sub and :unsub methods, the key will be passed as "channel".
b. Passing a value
If the triggering element has a value attribute (e.g., input elements or buttons with a manually set value), the value will be available as req["value"]. For example:
<button onclick=@update-theme value="dark">Dark mode</button>
Clicking the button will trigger the :update method with "theme" as the key and the "dark" as the value.
3. Triggering a Publish
The :pub method can be triggered from Tamnoon HEEx by using the special @pub syntax:
@pub-<channel>-<method>-<key>
	<channel>: The name of the channel to publish to.

	<method>: The method to invoke on all clients in that channel.

	<key> (optional): A key to pass along with the method call.


Example
The following button will broadcast the :ping method to all clients in the "clients" channel when clicked:
<button onclick=@pub-clients-ping>Ping everyone!</button>
This is equivalent to calling the :pub method in a method with:
Tamnoon.Methods.pub(%{
  "channel" => "clients",
  "action" => %{"method" => "ping"}
}, state)
Multiple triggers in an element
An element can have multiple event handlers for the same event in Tamnoon, and all of them will be triggered when the event fires.
However, this is usually not recommended and can often be replaced with a cleaner, combined handler.
Example
<button onclick=@first_method onclick=@second_method>Activate both methods!</button>




  

    
DOM Actions
    

DOM Actions provide targeted, low-level manipulation of the DOM without requiring interaction with the state. They closely resemble the browser's native DOM API, offering fine-grained control and serving as an escape hatch from Tamnoon's state-driven model.
While Tamnoon's state model is suited for most use cases, some scenarios benefit from direct DOM manipulation. DOM Actions are especially useful in the following cases:
Lists
Actions such as Tamnoon.DOM.Actions.AddChild and Tamnoon.DOM.Actions.RemoveNode enable you to manage long or dynamic lists efficiently. This avoids the overhead of updating and re-rendering the entire list through state changes.
Lazy Inputs
Inputs in Tamnoon are typically bound using value, meaning each change triggers a state sync with the server. This can introduce latency during typing. Instead, you can remove the value assignment and use Tamnoon.DOM.Actions.SetValue to manually change the input's value when needed. This approach improves responsiveness while still enabling programmatic control.
Minimal State
State fields can accumulate values used in only one place - for example, toggling the visibility of a button. Using actions like Tamnoon.DOM.Actions.ToggleAttribute, you can handle such logic directly in the DOM, keeping your state clean and focused only on the data that matters.

  
    
  
  In Practice


To execute a DOM action, it must first be constructed and then returned from a method. DOM actions are a specialized kind of DOM struct - data structures that resemble an abstract syntax tree (AST) and encode what to do, to which elements, and how.

  
    
  
  Constructing a DOM Action


To construct a DOM action, call its new!/1 function and pass in a map with the required arguments. These arguments can be literals (e.g., strings, booleans) or other DOM structs (such as nodes or node collections).
Each action has its own set of required fields - refer to its new!/1 documentation for the specific structure.

  
    
  
  DOM Structs


DOM structs are structured representations of operations and elements in the DOM. They are the building blocks for DOM actions, and they fall into one of the following categories:
	Action - Represents what to do. For example: Tamnoon.DOM.Actions.RemoveNode removes a node from the DOM.

	Node - Represents a single DOM node, i.e. what the action is applied to. See Tamnoon.DOM.Node.

	Node Collection Represents a group of DOM elements that are targets of the action. See Tamnoon.DOM.NodeCollection.


Example
The following method clears the input field with the ID "message-input":
defmethod :clear_message_input do
  input_node = Tamnoon.DOM.Node.new!(%{
    selector_type: :id,
    selector_value: "message-input"
  })

  clear_action = Tamnoon.DOM.Actions.SetValue.new!(%{
    target: input_node,
    value: ""
  })

  {%{}, [clear_action]}
end
In this example:
	A DOM node is constructed to target the input with ID "message-input".

	A SetValue action is created to set its value to an empty string.

	The action is returned from the method (alongside an empty diffs map), and will be executed on the client.



  
    
  
  Node Collections & ForEach


To apply actions to multiple DOM elements, you can use a node collection together with the Tamnoon.DOM.Actions.ForEach action.
Unlike most actions, ForEach is special in that it:
	Accepts another action as a parameter (the :callback parameter)

	Uses a special iteration placeholder as the target inside that callback. This placeholder represents each node in the collection as it's iterated.


Example
Tamnoon.DOM.Actions.ForEach.new!(%{
  target:
    Tamnoon.DOM.NodeCollection.new!(%{
      selector_type: :query,
      selector_value: ".toggle-me"
    }),
  callback:
    Tamnoon.DOM.Actions.ToggleAttribute.new!(%{
      target:
        Tamnoon.DOM.Node.new!(%{
          selector_type: :iteration_placeholder,
          selector_value: nil
        }),
        attribute: "hidden"
    })
})
In this example:
	The :target is a node collection selecting all elements with the "toggle-me" class.

	The :callback is a ToggleAttribute action that:
	Targets each node in the collection (via the :iteration_placeholder selector).

	Toggles its "hidden" attribute.




This results in every ".toggle-me" element being shown or hidden depending on its current state.
The Iteration Placeholder
The iteration placeholder represents a node that will be dynamically replaced with each element from the :target node collection during a ForEach operation.
As shown in the example, an iteration placeholder node is constructed as follows:
Tamnoon.DOM.Node.new!(%{
  selector_type: :iteration_placeholder,
  selector_value: nil
})
Note: The iteration placeholder is only valid within the context of a ForEach action. It is not intended for use outside of that context and will not function correctly elsewhere.




  

    
Tamnoon 
    



      
This module provides functions needed to initialize Tamnoon. You do not need to handle
it directly, rather, the only time you need to call something in this module is in your
supervision tree, to add it to the children and configure it (see child_spec/1).

  
    
  
  Example


def start_link(opts \\ []) do
  children = [Tamnoon]
  opts = [strategy: :one_for_one, name: Tamnoon.Supervisor]
  Supervisor.start_link(children, opts)
end

    Tamnoon.Compiler - Tamnoon v1.0.0-rc.1
    
    

    



  
  

    
Tamnoon.Compiler 
    



      
This module provides functions for compiling (/parsing) Tamnoon HEEx components.

    Tamnoon.Component - Tamnoon v1.0.0-rc.1
    
    

    



  
  

    
Tamnoon.Component behaviour
    



      
This module defines the Tamnoon.Component behaviour, which includes the
heex/0 callback. It should be implemented in every component module.

Tamnoon.MethodManager 
    



      
This module handles the management of different methods as you create them.
Notably, it provides the defmethod/2 macro.
Importing the module
In order to create handlers for the methods you set up, you must import Tamnoon.MethodManager
in your methods module. Then, you can use the defmethod/2 macro to implement handling
of the methods.

Tamnoon.Methods 
    



      
A methods module that provides built-in methods for use in your Tamnoon application.
This module offers utilities for basic state management via tmnn_get/2 and tmnn_update/2,
as well as PubSub functionality through tmnn_sub/2, tmnn_unsub/2, tmnn_pub/2 and subbed_channels/0.
Most of these methods are primarily intended either to be used internally by Tamnoon,
or to be triggered from Tamnoon HEEx markup.
This module is automatically included as a methods module by Tamnoon.
Note: In the documentation for the built-in methods, the term "returns" may refer to
either the function's return value or the value included in the server's response to the
client. The context should clarify which is meant.

Tamnoon.Router 
    



      
This module provides a default router for HTTP(S) requests, which builds and serves the web app.

Tamnoon.SocketHandler 
    



      
Tamnoon's :cowboy_websocket implementation. There should be no reason to interact with this
module, however, it is still documented to allow extensibility.
You can replace the module with another by setting the :socket_handler value in
Tamnoon.start_link/1's options.

Tamnoon.DOM.Node 
    



      
Represents a DOM node. Accepts two parameters for how to select the node,
:selector_type and :selector_value. For example, when :selector_type is :id and
:selector_value is "id-to-select", it will be equivalent to using
document.getElementById("id-to-select") in JS.
:selector_type can be set as :iteration_placeholder, which is a special value intended for use
with Tamnoon.DOM.Actions.ForEach, and will represent the argument that will be each node
of the collection.

  
    
  
  Fields


	:selector_type – an atom indicating the type of selector.
	Expected values: :id, :xpath, :from_string, :first_element, :last_element, :iteration_placeholder.


	:selector_value – the value associated with the selector.
	:id: a String.t/0 of the id to select.
	:xpath: a String.t/0 of the node's xpath.
	:from_string: a String.t/0 representing the HTML element.
	:first_element and :last_element: a Tamnoon.DOM.NodeCollection.t/0 representing the collection to query.
	:iteration_placeholder: nil.

Tamnoon.DOM.NodeCollection 
    



      
Represents a DOM node collection. Accepts two parameters for how to select the nodes,
:selector_type and :selector_value. For example, when :selector_type is :children and
:selector_value is a Tamnoon.DOM.Node.t/0, it will be equivalent to using
element.children() in JS.

  
    
  
  Fields


	:selector_type – an atom indicating the type of selector.
	Expected values: :xpath, :query, :children.


	:selector_value – the value associated with the selector.
	:xpath: a String.t/0 of the xpath to select by.
	:query: a String.t/0 of the query selector to select by.
	:children: a Tamnoon.DOM.Node.t/0 of the parent node.

Tamnoon.DOM.Actions.AddChild 
    



      
An action that will append the :child node to the :parent node's children.

Tamnoon.DOM.Actions.ForEach 
    



      
An action that will trigger the :callback action on each element in the
collection :target.

Tamnoon.DOM.Actions.RemoveNode 
    



      
An action that will remove the :target node from the DOM.

Tamnoon.DOM.Actions.ReplaceNode 
    



      
An action that will replace the :target node with the :replacement node.

      


      
        Summary


  
    Types
  


    
      
        
  
    
    Tamnoon.DOM.Actions.SetAttribute - Tamnoon v1.0.0-rc.1
    
    

    



  
  

    
Tamnoon.DOM.Actions.SetAttribute 
    



      
An action that will set the :attribute attribute of the :target node
to :value. Can also be used to set the textContent of a node by passing "textContent"
to :attribute.

Tamnoon.DOM.Actions.SetInnerHTML 
    



      
An action that will replace the :target node's innerHTML with :value.

